# $Id$
# Maintainer: Eric Bélanger <eric@archlinux.org>

pkgname=lcms2
pkgver=2.9
pkgrel=1
pkgdesc="Small-footprint color management engine, version 2"
arch=('x86_64')
url="http://www.littlecms.com"
license=('MIT')
depends=('libtiff')
source=(http://downloads.sourceforge.net/sourceforge/lcms/${pkgname}-${pkgver}.tar.gz)
sha1sums=('60bea9875e017dd1c466e988c2ad98f8766e4e55')

prepare() {
  cd $pkgname-$pkgver
}

build() {
  cd ${pkgname}-${pkgver}
  ./configure --prefix=/usr
  make
}

check() {
  cd ${pkgname}-${pkgver}
  make check
}

package() {
  cd ${pkgname}-${pkgver}
  make DESTDIR="${pkgdir}" install
  install -Dm644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/LICENSE"
}
